SOLCH - A FORTRAN PACKAGE FOR SOLUTIONS OF PERTURBED LINEAR EQUATIONS

J.W. Bandler and Q.J. Zhang

SOS-84-21-L

December 1984

SOLCH - A FORTRAN PACKAGE FOR SOLUTIONS OF PERTURBED LINEAR EQUATIONS

J.W. Bandler and Q.J. Zhang

Abstract

SOLCH is a package of subroutines for evaluating solutions of perturbed linear equations. Appropriate large change formulas are used to ensure efficient computation. The package and documentation have been developed for use on the CDC 170/815 system with the NOS 2.2-602/587 operating system and the Fortran Extended (FTN) version 4.8 compiler. This document contains a listing of the SOLCH package.

This work was supported in part by the Natural Sciences and Engineering Research Council of Canada under Grant A7239.

The authors are with the Simulation Optimization Systems Research Laboratory and the Department of Electrical and Computer Engineering, McMaster University, Hamilton, Canada L8S 4L7.

I. INTRODUCTION

SOLCH is a package of Fortran subroutines for solving a perturbed system of linear equations. It uses large change formulas (47) and (48) in our previous report [1] and can essentially reduce the computation from solving an nxn system to solving an rxr smaller system, where r is the rank of the coefficient deviation matrix of the linear equations.

The whole package is written in Fortran IV for the CDC 170/815 system with the NOS 2.2-602/587 operating system. It is available at McMaster University in the form of a library of binary relocatable subroutines in the group indirect file LIBSLCH under the charge RJWBAND.

This document includes a listing of the package SOLCH. The user's manual presented together with illustrative examples is found in [2]. The listing contains 372 lines, of which 204 are comments. It has been modularized into 5 subroutines. The list of all subroutines is given in Table I.

TABLE I

LIST OF SUBROUTINES OF THE SOLCH PACKAGE

	Subroutine	Number of Lines (source text)	Listing from Page
1	SOLCH	70	4
2	SOLU0	91	5
3	SOLCHA	51	6
4	SOLCHB	91	7
5	LUFACT	62	8

II. REFERENCES

[1] J.W. Bandler and Q.J. Zhang, "A unified approach to first-order and large change sensitivity computations in linear systems", Department of Electrical and Computer Engineering, McMaster University, Canada, Report SOS-84-20-R, 1984.

[2] J.W. Bandler and Q.J. Zhang, "SOLCH - A Fortran package for solutions of perturbed linear equations", Department of Electrical and Computer Engineering, McMaster University, Canada, Report SOS-84-21-U, 1984.

III. LISTING OF THE SOLCH PACKAGE

```
      SUBROUTINE SOLCH(N,LW0,A,B,W0,ICH)                                    000001
C                                                                           000002
C     A MAN-MACHINE INTERACTIVE PROGRAM TO CALCULATE PERTURBED SOLUTIONS    000003
C     OF LINEAR EQUATIONS.  LARGE CHANGE FORMULAS (47) AND (48) OF          000004
C     REFERENCE [1] ARE USED.                                               000005
C                                                                           000006
C     THE LINEAR SYSTEM IS  [A][X]=[B].                                     000007
C     THE PERTURBED SYSTEM IS  ( [A]+[V][D][W]' )[X]=[B].                   000008
C                                                                           000009
C     LIST OF ARGUMENTS : ( ....  FOR INPUT ARGUMENTS.                      000010
C                         ----  FOR OUTPUT ARGUMENTS. )                    000011
C      N     ....  NO. OF ROWS OR COLUMNS OF [A].                           000012
C      LW0   ....  AN INTEGER DEFINING THE DIMENSION OF ARRAY W0.           000013
C                  RECOMMENDED VALUE IS N*(4+6*N).  BUT IF IR1 OR IR2       000014
C                  IS MUCH LESS THAN N, THEN LW0 CAN BE GREATLY REDUCED.    000015
C                  PRECISELY, LW0=N*(N+3)+2*IR1*(N+IR2)+IR2*(1+N). WHERE    000016
C                  IR1 AND IR2 ARE THE NO. OF ROWS AND COLUMNS OF [D]       000017
C                  RESPECTIVELY.                                            000018
C      A     ..--  REAL ARRAY OF DIMENSION N BY N. ON ENTRY, IT MUST        000019
C                  BE SET TO [A]. ON EXIT, IT CONTAINS THE LU FACTORS       000020
C                  OF [A].                                                  000021
C      B     ..--  REAL ARRAY OF DIMENSION N. ON ENTRY, IT MUST BE          000022
C                  SET TO [B], THE R.H.S. OF THE LINEAR EQUATIONS.          000023
C                  ON EXIT, IT CONTAINS [X0], THE SOLUTON OF THE            000024
C                  EQUATIONS BEFORE LARGE CHANGES.                          000025
C      W0    ----  REAL ARRAY OF DIMENSION LW0, USED FOR WORKING SPACES.    000026
C      ICH   ....  AN INTEGER DEFINING THE CHANNEL NUMBER FOR OUTPUT        000027
C                  PRINT-OUT.                                               000028
C                                                                           000029
      DIMENSION A(N,N),B(N),W0(LW0)                                         000030
C                                                                           000031
      PRINT*,"*************************************************************" 000032
      PRINT*,"SOLVING ([A]+[V][D][W]')[X]=[B] BY LARGE CHANGE FORMULA"      000033
      PRINT*," "                                                            000034
C                                                                           000035
C     PERFORM LU FACTORIZATION AND FORWARD AND BACKWARD SUBSTITUTION         000036
C     TO SOLVE THE ORIGINAL SYSTEM.                                         000037
C                                                                           000038
      CALL LUFACT(N,A,B,1)                                                  000039
      PRINT*,"   SOLUTIONS BEFORE LARGE CHANGE : "                          000040
      WRITE(ICH,50) B                                                       000041
C                                                                           000042
   10 PRINT*," ENTER  IR1 AND IR2   ( SUCH THAT [D] IS OF IR1 BY IR2 )"     000043
      READ*,IR1,IR2                                                         000044
C                                                                           000045
C     CALCULATE RELEVANT PARAMETERS FOR MEMORY SPACE DISTRIBUTION.          000046
C                                                                           000047
      N2=1+N                                                                000048
      N3=N2+N*IR1                                                           000049
      N4=N3+IR1*IR2                                                         000050
      N5=N4+N*IR2                                                           000051
      N6=N5+N*IR1                                                           000052
      N7=N6+IR1*IR2                                                         000053
      N8=N7+IR2                                                             000054
      N9=N8+N*N                                                             000055
      N10=N9+N                                                              000056
C                                                                           000057
C     CALL THE MAIN SUBROUTINE TO SOLVE THE SYSTEM EQUATION AFTER           000058
C     CHANGE.  FORMULA (47) OR (48) OF REF. [1] IS USED .                   000059
C                                                                           000060
      CALL SOLU0(N,IR1,IR2,A,B,W0(1),W0(N2),W0(N3),W0(N4),W0(N5),           000061
     +W0(N6),W0(N7),W0(N8),W0(N9),W0(N10),ICH)                              000062
C                                                                           000063
      PRINT*," TRY ANOTHER SET OF [V], [D] AND [W] ?   "                    000064
      PRINT*,"  ENTER 1 OR 2  ( Y OR N ) "                                  000065
```

```
      READ*,KKK                                                          000066
      IF(KKK.EQ.1) GO TO 10                                              000067
      RETURN                                                             000068
   50 FORMAT(//31X,"VECTOR [X]"//10(22X,F18.5//)/)                       000069
      END                                                                000070
C                                                                        000071
C                                                                        000072
      SUBROUTINE SOLU0(N,IR1,IR2,A,X0,X,V,D,W,PV,WAV,RHS,H,W1,W2,ICH)    000073
C                                                                        000074
C     FORMULATE A MAN-MACHINE INTERACTIVE PROCEDURE TO SOLVE             000075
C     ( [A]+[V][D][W]' )[X]=[B].                                         000076
C                                                                        000077
C     LIST OF ARGUMENTS :  (  ....  FOR INPUT ARGUMENTS.                 000078
C                             ----  FOR OUTPUT ARGUMENTS. )              000079
C                                                                        000080
C      N      ....  NO. OF ROWS OR COLUMNS OF [A].                       000081
C      IR1    ....  NO. OF ROWS OF [D].                                  000082
C      IR2    ....  NO. OF COLUMNS OF [D].                               000083
C      A      ....  REAL ARRAY OF DIMENSION N BY N, CONTAINING  THE      000084
C                   LU FACTORS OF [A] IN COMPACT FORM.                   000085
C      X0     ....  REAL ARRAY OF DIMENSION N, CONTAINING THE SOLUTION   000086
C                   OF THE SYSTEM BEFORE LARGE CHANGE.                   000087
C      X      ----  REAL ARRAY OF DIMENSION N, CONTAINING THE SOLUTION   000088
C                   OF THE SYSTEM AFTER LARGE CHANGE.                    000089
C      V,D,W ----   REAL ARRAYS OF DIMENSIONS N BY IR1, IR1 BY IR2 AND   000090
C                   N BY IR2 RESPECTIVELY,SUCH THAT                      000091
C                      DELTA ( [A] )=[V][D][W]'.                         000092
C      PV,WAV,RHS,H,W1,W2  ----  REAL ARRAYS OF DIMENSIONS N BY IR1,      000093
C                   IR2 BY IR1,IR2, N BY N, N AND N RESPECTIVELY,        000094
C                   USED AS WORKING SPACES.                              000095
C                                                                        000096
      DIMENSION A(N,N),X0(N),X(N),V(N,IR1),D(IR1,IR2),W(N,IR2)           000097
      DIMENSION PV(N,IR1),WAV(IR2,IR1),RHS(IR2),H(N,N)                   000098
      DIMENSION W1(N),W2(N)                                              000099
C                                                                        000100
      PRINT*," [V], [D] AND [W] ARE MATRICES OF DIMENSIONS N BY IR1,"    000101
      PRINT*,"                  IR1 BY IR2 AND N BY IR2 RESPECTIVELY"    000102
      PRINT*,"                  SUCH THAT  DELTA( [A] )=[V][D][W]'"      000103
      PRINT*," "                                                         000104
      PRINT*,"ENTER [V]            ( COLUMN BY COLUMN )"                 000105
      READ*,V                                                            000106
      PRINT*,"ENTER [W]            ( COLUMN BY COLUMN )"                 000107
      READ*,W                                                            000108
C                                                                        000109
      WRITE(ICH,70) N,IR1,IR2                                            000110
      PRINT*,"     MATRIX [V] : "                                        000111
      DO 3 I=1,N                                                         000112
    3 WRITE(ICH,90) (V(I,J),J=1,IR1)                                     000113
      WRITE(ICH,100)                                                     000114
      PRINT*,"     MATRIX [W] : "                                        000115
      DO 5 I=1,N                                                         000116
    5 WRITE(ICH,90) (W(I,J),J=1,IR2)                                     000117
      WRITE(ICH,100)                                                     000118
C                                                                        000119
C     CALL SUBROUTINE SOLCHA TO SOLVE [A][PV]=[V] FOR [PV],CALCULATE     000120
C     [WAV]=[W]'[PV]  AND [RHS]=[W]'[X0].  THESE CALCULATIONS ARE        000121
C     PERFORMED ONLY ONCE FOR A PARTICULAR SET OF [V] AND [W].           000122
C                                                                        000123
      CALL SOLCHA(N,IR1,IR2,A,X0,V,W,PV,WAV,RHS,W1)                      000124
C                                                                        000125
      KKK=0                                                              000126
   10 PRINT*,"[D] IS OF ",IR1," BY ",IR2,"."                             000127
      PRINT*,"ENTER [D]            ( COLUMN BY COLUMN )"                 000128
      READ*,D                                                            000129
      PRINT*," "                                                         000130
```

```
C                                                                       000131
C     IF KKK IS NON-ZERO, THEN [V] AND [W] ARE STILL THE SAME AS THAT   000132
C     WITH THE PREVIOUS [D].                                            000133
C                                                                       000134
      IF(KKK.NE.0) WRITE(ICH,80)                                        000135
      PRINT*,"      MATRIX [D] : "                                      000136
      DO 20 I=1,IR1                                                     000137
   20 WRITE(ICH,90) (D(I,J),J=1,IR2)                                    000138
      WRITE(ICH,100)                                                    000139
C                                                                       000140
C     CALL SUBROUTINE SOLCHB TO CALCULATE [X] USING FORMULA (47) OR     000141
C     (48) OF REF. [1].                                                 000142
C                                                                       000143
      IR=IR2                                                            000144
      IF(IR1.LT.IR2) IR=IR1                                             000145
      CALL SOLCHB(N,IR1,IR2,IR,D,PV,WAV,RHS,X0,X,H,W1,W2)               000146
      PRINT*,"   SOLUTIONS AFTER LARGE CHANGE : "                       000147
      WRITE(ICH,50) X                                                   000148
C                                                                       000149
      PRINT*," TRY ANOTHER [D] ?   ENTER 1 OR 2  ( Y OR N )    "        000150
      READ*,KKK                                                         000151
      IF(KKK.EQ.2) RETURN                                               000152
      GO TO 10                                                          000153
   50 FORMAT(//31X,"VECTOR [X]"//10(22X,F18.5//)/)                      000154
   70 FORMAT(1H1,                                                       000155
     +4X,"NUMBER OF ROWS OR COLUMNS OF [A] ( N )  .........",I3//       000156
     +5X,"NUMBER OF ROWS OF [D] ( IR1 )           .........",I3//       000157
     +5X,"NUMBER OF COLUMNS OF [D] ( IR2 )        .........",I3//)      000158
   80 FORMAT(1H1,5X,"[V] AND [W] ARE THE SAME AS THOSE WITH THE",       000159
     +" PREVIOUS [D]."//)                                               000160
   90 FORMAT(/10X,10F10.5/)                                             000161
  100 FORMAT(//)                                                        000162
      END                                                               000163
C                                                                       000164
C                                                                       000165
      SUBROUTINE SOLCHA(N,IR1,IR2,A,X0,V,W,PV,WAV,RHS,W1)               000166
C                                                                       000167
C     SOLVE [A][PV]=[V] AND CALCULATE [WAV]=[W]'[PV] AND [RHS]=[W]'[X0]. 000168
C     THIS SUBROUTINE PROVIDES COMMON FACTORS [PV],[WAV] AND [RHS] FOR  000169
C     DIFFERENT [D] WITH THE SAME SET OF [V] AND [W].                   000170
C                                                                       000171
C     LIST OF ARGUMENTS : (  ....  FOR INPUT ARGUMENTS.                 000172
C                            ----  FOR OUTPUT ARGUMENTS. )              000173
C      N     ....  NO. OF ROWS OR COLUMNS OF [A].                       000174
C      IR1   ....  NO. OF ROWS OF [D].                                  000175
C      IR2   ....  NO. OF COLUMNS OF [D].                               000176
C      A     ....  REAL ARRAY OF DIMENSION N BY N, CONTAINING THE       000177
C                  LU FACTORS OF [A] IN COMPACT FORM.                   000178
C      X0    ....  REAL ARRAY OF DIMENSION N, CONTAINING THE SOLUTION   000179
C                  OF [A][X0]=[B], I.E., SOLUTION BEFORE CHANGE.        000180
C      V     ....  REAL ARRAY OF DIMENSION N BY IR1, CONTAINING [V].    000181
C      W     ....  REAL ARRAY OF DIMENSION N BY IR2, CONTAINING [W].    000182
C      PV    ----  REAL ARRAY OF DIMENSION N BY IR1, CONTAINING [PV]    000183
C                  WHICH IS THE SOLUTION OF [A][PV]=[V].                000184
C      WAV   ----  REAL ARRAY OF DIMENSION IR2 BY IR1, CONTAINING THE   000185
C                  MATRIX PRODUCT OF [W]'[PV].                          000186
C      RHS   ----  REAL ARRAY OF DIMENSION IR2, CONTAINIG THE PRODUCT   000187
C                  OF [W]'[X0].                                         000188
C      W1    ----  REAL ARRAY OF DIMENSION N, USED AS WORKING SPACES.   000189
C                                                                       000190
      DIMENSION A(N,N),V(N,IR1),W(N,IR2),PV(N,IR1),RHS(IR2)             000191
      DIMENSION WAV(IR2,IR1),X0(N),W1(N)                                000192
C                                                                       000193
C     PERFORM FBS ( FORWARD AND BACKWARD SUBSTITUTION ) FOR THE         000194
C     SOLUTION OF [A][PV]=[V].                                          000195
```

```
C                                                                       000196
      DO 20 I=1,IR1                                                     000197
      DO 10 J=1,N                                                       000198
   10 W1(J)=V(J,I)                                                      000199
      CALL LUFACT(N,A,W1,3)                                             000200
      DO 20 J=1,N                                                       000201
   20 PV(J,I)=W1(J)                                                     000202
C                                                                       000203
C     PERFORM MATRIX AND VECTOR MULTIPLICATIONS FOR [W]'[PV] AND        000204
C     [W]'[X0].  RESULTS ARE STORED IN ARRAYS WAV AND RHS RESPECTIVELY. 000205
C                                                                       000206
      DO 40 I=1,IR2                                                     000207
      DO 30 J=1,IR1                                                     000208
      WAV(I,J)=0.0                                                      000209
      DO 30 K=1,N                                                       000210
   30 WAV(I,J)=WAV(I,J)+W(K,I)*PV(K,J)                                  000211
      RHS(I)=0.                                                         000212
      DO 40 K=1,N                                                       000213
   40 RHS(I)=RHS(I)+W(K,I)*X0(K)                                        000214
      RETURN                                                            000215
      END                                                               000216
C                                                                       000217
C                                                                       000218
      SUBROUTINE SOLCHB(N,IR1,IR2,IR,D,PV,WAV,RHS,X0,X,H,W1,S)          000219
C                                                                       000220
C     USE LARGE CHANGE FORMULA (47) OR (48) OF REF. [1] TO SOLVE        000221
C     ( [A]+[V][D][W]' )[X]=[B].  SUBROUTINE "SOLCHA" SHOULD BE         000222
C     EXECUTED AT LEAST ONCE TO PROVIDE [PV],[WAV] AND [RHS]. SEE       000223
C     THE ARGUMENT LIST FOR DETAILS.                                    000224
C                                                                       000225
C     LIST OF ARGUMENTS : (  ....  FOR INPUT ARGUMENTS.                 000226
C                            ----  FOR OUTPUT ARGUMENTS. )              000227
C      N     ....  NO. OF ROWS OR COLUMNS OF [A].                       000228
C      IR1   ....  NO. OF ROWS OF [D].                                  000229
C      IR2   ....  NO. OF COLUMNS OF [D].                               000230
C      IR    ....  EQUALS IR1 IF IR1 < IR2, OTHERWISE EQUALS IR2.       000231
C      D     ....  REAL ARRAY OF DIMENSION IR1 BY IR2, CONTAINING [D].  000232
C      PV    ....  REAL ARRAY OF DIMENSION N BY IR1, CONTAINING THE     000233
C                  SOLUTION OF [A][PV]=[V].                             000234
C      WAV   ....  REAL ARRAY OF DIMENSION IR2 BY IR1, CONTAINING THE   000235
C                  MATRIX PRODUCT OF [W]'[PV].                          000236
C      RHS   ....  REAL ARRAY OF DIMENSION IR2, CONTAINING THE PRODUCT OF 000237
C                  [W]'[X0].                                            000238
C      X0    ....  REAL ARRAY OF DIMENSION N, CONTAINING THE SOLUTION OF 000239
C                  [A][X0]=[B], I.E., THE SOLUTION BEFORE CHANGE.       000240
C      X     ----  REAL ARRAY OF DIMENSION N, CONTAINING THE SOLUTION OF 000241
C                  ( [A]+[V][D][W]' )[X]=[B], I.E., THE SOLUTION OF THE 000242
C                  SYSTEM AFTER CHANGE.                                 000243
C      H,W1,S --   REAL ARRAYS OF DIMENSIONS IR BY IR, N AND IR         000244
C                  RESPECTIVELY, USED AS WORKING SPACES.                000245
C                                                                       000246
      DIMENSION D(IR1,IR2),PV(N,IR1),WAV(IR2,IR1),RHS(IR2),H(IR,IR)     000247
      DIMENSION X0(N),X(N),W1(N),S(IR)                                  000248
C                                                                       000249
C     IF IR1 IS SMALLER THAN IR2, THEN FORMULA (48) OF REF. [1]         000250
C     IS USED, OTHERWISE FORMULA (47) IS USED.                          000251
C                                                                       000252
      IF(IR1.LT.IR2) GO TO 50                                           000253
C                                                                       000254
C     FORMULATE THE COEFFICIENT MATRIX OF THE IR2 BY IR2 "SMALLER       000255
C     SYSTEM" . THE RESULTS ARE STORED IN ARRAY H.                      000256
C                                                                       000257
      DO 20 I=1,IR2                                                     000258
      S(I)=RHS(I)                                                       000259
      DO 20 J=1,IR2                                                     000260
```

```
      H(I,J)=0.                                                          000261
      IF(I.EQ.J) H(I,J)=1.                                               000262
      DO 20 K=1,IR1                                                      000263
   20 H(I,J)=H(I,J)+WAV(I,K)*D(K,J)                                      000264
C                                                                        000265
C     SOLVE THE "SMALLER SYSTEM" : [H][S]=[RHS].                         000266
C                                                                        000267
      CALL LUFACT(IR2,H,S,1)                                             000268
C                                                                        000269
C     CALCULATE THE CORRESPONDING SOLUTIONS OF THE SYSTEM AFTER CHANGE.  000270
C     USING THE EXPRESSION [X]=[X0]-[PV][D][S].                          000271
C                                                                        000272
      DO 30 I=1,IR1                                                      000273
      W1(I)=0.                                                           000274
      DO 30 J=1,IR2                                                      000275
   30 W1(I)=W1(I)+D(I,J)*S(J)                                            000276
C                                                                        000277
      DO 40 I=1,N                                                        000278
      X(I)=X0(I)                                                         000279
      DO 40 J=1,IR1                                                      000280
   40 X(I)=X(I)-PV(I,J)*W1(J)                                            000281
      RETURN                                                             000282
C                                                                        000283
C     FORMULATE THE COEFFICIENT MATRIX OF THE IR1 BY IR1 " SMALLER       000284
C     SYSTEM ".  THE RESULTS ARE STORED IN H.                            000285
C                                                                        000286
   50 DO 70 I=1,IR1                                                      000287
      S(I)=0.                                                            000288
      DO 60 J=1,IR2                                                      000289
   60 S(I)=S(I)+D(I,J)*RHS(J)                                            000290
      DO 70 J=1,IR1                                                      000291
      H(I,J)=0.                                                          000292
      IF(I.EQ.J) H(I,J)=1.                                               000293
      DO 70 K=1,IR2                                                      000294
   70 H(I,J)=H(I,J)+D(I,K)*WAV(K,J)                                      000295
C                                                                        000296
C     SOLVE THE " SMALLER SYSTEM " :  [H][S]=[D][W]'[X0].                000297
C                                                                        000298
      CALL LUFACT(IR1,H,S,1)                                             000299
C                                                                        000300
C     CALCULATING CORRESPONDING SOLUTION VECTOR USING                    000301
C         [X]=[X0]-[PV][S].                                              000302
C                                                                        000303
      DO 80 I=1,N                                                        000304
      X(I)=X0(I)                                                         000305
      DO 80 J=1,IR1                                                      000306
   80 X(I)=X(I)-PV(I,J)*S(J)                                             000307
      RETURN                                                             000308
      END                                                                000309
C                                                                        000310
      SUBROUTINE LUFACT(N,A,B,MODE)                                      000311
C                                                                        000312
C     PERFORM LU-FACTORIZATION AND/OR FORWARD AND BACKWARD               000313
C     SUBSTITUTION ( FBS ) FOR THE SOLUTION OF [A][X]=[B].               000314
C                                                                        000315
C     LIST OF ARGUMENTS :  (  .... FOR INPUT ARGUMENTS.                  000316
C                             ---- FOR OUTPUT ARGUMENTS.)                000317
C      N     ....  NO. OF ROWS OR COLUMNS OF [A].                        000318
C      A     ..--  REAL ARRAY OF DIMENSION N BY N. ON ENTRY, IT SHOULD    000319
C                  BE SET TO [A] WHEN MODE=1 OR, SET TO THE LU FACTORS    000320
C                  OF [A] IF MODE=3.  ON EXIT, IT CONTAINS THE LU FACTORS 000321
C                  OF [A].                                                000322
C      B     ..--  REAL ARRAY OF DIMENSION N.  ON ENTRY, IT CONTAINS     000323
C                  THE R.H.S. OF THE LINEAR EQUATIONS, I.E, B. ON         000324
C                  EXIT, IT CONTAINS THE SOLUTION VECTOR [X].             000325
```

```
C       MODE ....  =1,  LU-FACTORIZATION AND FBS.                        000326
C                  =3,  FBS ONLY.                                        000327
C                                                                        000328
C                                                                        000329
      DIMENSION A(N,N),B(N)                                              000330
C                                                                        000331
      N2=N-1                                                             000332
      N1=N+1                                                             000333
      IF(MODE.EQ.3.OR.N.EQ.1) GO TO 50                                   000334
C                                                                        000335
C     CROUT'S ALGORITHM OF LU-FACTORIZATION.                             000336
C                                                                        000337
      DO 20 II=1,N2                                                      000338
      ZZ=A(II,II)                                                        000339
      IF(ABS(ZZ).LT.1.E-20) GO TO 100                                    000340
      NN=II+1                                                            000341
      DO 20 J=NN,N                                                       000342
      A(II,J)=A(II,J)/ZZ                                                 000343
      DO 10 K=NN,N                                                       000344
   10 A(K,J)=A(K,J)-A(K,II)*A(II,J)                                      000345
   20 CONTINUE                                                           000346
C                                                                        000347
C     FORWARD SUBSTITUTION                                               000348
C                                                                        000349
   50 IF(ABS(A(1,1)).LT.1.E-20) GO TO 100                                000350
      B(1)=B(1)/A(1,1)                                                   000351
      IF(N.EQ.1) RETURN                                                  000352
      DO 70 II=2,N                                                       000353
      ZZ=A(II,II)                                                        000354
      IF(ABS(ZZ).LT.1.E-20) GO TO 100                                    000355
      IN=II-1                                                            000356
      DO 60 K=1,IN                                                       000357
   60 B(II)=B(II)-A(II,K)*B(K)                                           000358
   70 B(II)=B(II)/ZZ                                                     000359
C                                                                        000360
C     BACKWARD SUBSTITUTION                                              000361
C                                                                        000362
      DO 80 L=2,N                                                        000363
      II=N1-L                                                            000364
      IP=II+1                                                            000365
      DO 80 K=IP,N                                                       000366
   80 B(II)=B(II)-A(II,K)*B(K)                                           000367
      RETURN                                                             000368
C                                                                        000369
  100 PRINT*,"INDEFINITE VALUE IN LUFACT DUE TO ZERO ON THE DIAGONAL"    000370
      RETURN                                                             000371
      END                                                                000372
```